
Véra Molnar et l’art optique

  L’Art optique rassemble des artistes fascinés par ce que l’œil perçoit du mouvement et de la 
lumière et qui en explorent les propriétés : intensité, rythme, variabilité, cycles... La démarche est bien sûr 
esthétique (non technique, physique ou métaphysique). C’est un mouvement riche et varié, qui s’exprime 
par des tableaux, des sculptures mais aussi des assemblages, installations ou montages vidéos. À ses débuts, 
on parle d’Art cinétique et optique. Dans les années 1960, l’appellation est simplifiée : on parle d’Art optique 
en France, d’Optique Art ou Op Art chez les anglo-saxons. 

 Véra Molnar est née en 1924 à Budapest (Hongrie). Depuis 1947, elle vit et travaille à Paris. Elle 
peut être présentée comme un peintre géométrique : les éléments de base de son travail sont parmi les plus 
simples, les plus élémentaires : la ligne,  le carré, le blanc, le noir, parfois des gris, des rouges, des bleus… 
A l’exploration de ces formes, elle a consacré des dizaines d’années ; et elle continue aujourd’hui. Elle a 
participé à tous les débats qui ont animé la naissance de l’art cinétique et permis la création de La Nouvelle 
Tendance et est devenue à partir de 1968 l’une des pionnières de l’utilisation de l’ordinateur dans la création 
artistique. Son art, conduit de façon expérimentale, porte sur la forme, sa transformation, son déplacement, 
sa perception. Son travail s’accompagne d’une intense réflexion théorique sur les moyens de la création et les 
mécanismes de la vision et trouve de nombreuses correspondances dans tous les travaux conduits en rapport 
avec les sciences exactes et les mathématiques en particulier.
 Le travail, qui, chez d’autres, pourrait être systématique voire “machinique”, a en réalité pour but de 
faire surgir l’imprévu, la liberté, l’imaginaire. Les lignes, par exemple, deviennent « extravagantes », comme 
nous le précisent d’ailleurs les titres de certains tableaux. (d’après la biographie de Véra Molnar présente sur le 
site de la galerie ONIRIS).

Cette œuvre s’inscrit dans une série consacrée aux quadrilatères autour de l’année 1984. Véra Molnar a conçu 
de nombreuses œuvres, toutes nommées Structure de quadrilatères, sur des supports différents : ordinateur, 
toile ...

Étude géométrique et algorithmique de l’œuvre

L’étude mathématique du tableau est indissociable de sa réalisation avec GéoTortue. Les propriétés géomé-
triques se dévoileront au fur et à mesure des problèmes rencontrés avec le logiciel.

Mise en place du problème

L’œuvre est composée de groupements de quadrilatères disposés en grille. Chaque groupement est composé 
de dix quadrilatères quelconques générés aléatoirement.

Il nous faut donc réaliser un quadrilatère quelconque avec GéoTortue. Pour cela, il faut chercher les liens qui 
existent entre les différentes dimensions du quadrilatères : les quatre longueurs et les quatres angles.

Après quelques essais de construction avec GéoTortue, nous constatons que les longueurs l1, l2 , l3 ainsi que les 
angles a1

t , a2
t , a3

t  sont choisis aléatoirement mais que les trois dernières dimensions l4, a3
t  et a4

t  sont liées aux 
autres. Il faut les déterminer afin de fermer le quadrilatère (l4, a3

t ) puis replacer la tortue dans sa position de 
départ ( a4

t  ).

Le problème revient donc à fermer le quadrilatère. Il peut être intéressant d’étudier ce problème de fermeture 
de figure dans le cas du triangle. Cela est en effet plus abordable pour les élèves et s’intègre parfaitement 
à notre recherche. Cela peut permettre aussi de (re)découvrir le théorème d’Al-Kashi dont l’utilisation est 
nécessaire pour avancer dans notre problème de quadrilatère. Cette démarche est détaillée dans la partie Avec 
les élèves.

Fermeture du quadrilatère

Il nous faut d’abord déterminer plusieurs mesures intermédiaires : la longueur d de la diagonale [BD], la 
mesure 1a  de l’angle ABD%  et la mesure 2a  de l’angle DBC% .

Pour la longueur BD, il suffit d’appliquer le théo-
rème d’Al-Kashi dans le triangle ABD.

De la même manière, l’angle ABD%  est donné par :

Nous en déduisons la mesure 2a  de l’angle DBC%  :

Nous pouvons désormais déterminer les mesures l4 et a3
t  du quadrilatère en appliquant le théorème d’Al-

Kashi dans le triangle DAC.

Enfin, nous pouvons déterminer la mesure a4
t  du quadrilatère.

Nous pouvons donc écrire une procédure quadrilatère avec GéoTortue. En voici une proposition.

pour quadrilatere l1 l2 l3 a1 a2
//variables intermédiaires
d:= √(l1²+l2²+2*l1*l2*cos(a1))
alpha1:= acos((l1²-l2²-d²)/(-2*l2*d))
alpha2:=180-(a2+alpha1)
l4:= √(d²+l3²-2*d*l3*cos(alpha2)) 
a3:= acos((d²-l3²-l4²)/(2*l3*l4))
a4:= 360-a1-a2-a3

//le tracé
av l1; td a1; av l2 ; td a2 ;av l3 ;td a3 ; av l4 ; td a4
fin

Alignement au centre de la tortue dans le quadrilatère

Pour générer des groupements de quadrilatères aléatoirement de la même manière que dans l’œuvre, il faut 
que ceux-ci soient centrés. Comme centre du quadrilatère, nous avons le choix entre l’intersection des dia-
gonales, l’isobarycentre ...
Il suffit d’essayer pour se rendre compte que l’intersection des diagonales n’est pas un choix approprié. En 
effet, atteindre le point d’intersection des diagonales depuis un sommet demande de nombreux calculs sup-
plémentaires. L’isobarycentre semble être le plus indiqué.

Ainsi, pour aller à l’isobarycentre, la tortue doit effectuer le trajet suivant :

Il nous faut donc déterminer les angles b  et c .

Partant du point D et arrivée au centre, la tor-
tue n’est pas orientée de la bonne manière. Il faut 
l’orienter dans la même position que lorsqu’elle est 
partie du point D. L’angle correspondant est l’angle 
ADC% , soit a180 4- t .

La procédure quadrilatère devient donc :            pour quadrilatere l1 l2 l3 a1 a2
//variables intermédiaires
d:= √(l1²+l2²+2*l1*l2*cos(a1))
alpha1:= acos((l1²-l2²-d²)/(-2*l2*d))
alpha2:=180-(a2+alpha1)
beta:=a1-alpha1
l4:= √(d²+l3²-2*d*l3*cos(alpha2)) 
a3:= acos((d²-l3²-l4²)/(2*l3*l4))
a4:= 360-a1-a2-a3
gamma:=180-a4-beta

//position du centre au sommet
lc
td (180-a4) ; re l4/4 ; tg gamma; re d/4 ; tg beta ; re l1/4
bc

//le tracé
av l1; td a1; av l2 ; td a2 ;av l3 ;td a3 ; av l4 ; td a4

//position du sommet au centre
lc
av l1/4 ; td beta ; av d/4 ; td gamma ;  av l4/4 ; td (180-a4) 
bc
fin

Création du motif

Le motif est constitué de dix quadrilatères que nous devons générer aléatoirement. Nous choisissons 
comme zone de délimitation approximative un carré de 200 de côté et empêchons que le quadrilatère soit 
trop petit.

Nous pouvons donc choisir les longueurs l1, l2 , l3 de manière à ce qu’elles soient comprises entre 50 et 200. 
Pour cela, nous utilisons la fonction rand() de GéoTortue.
Ainsi : 200 150 ()l l l rand1 2 3 #= = = - .

Les angles eux-aussi ne doivent pas être «trop» obtus, ni «trop» aigus. Nous pouvons les choisir afin qu’il 
soient compris entre 80 et 110. Ainsi : 110 30 ()a a rand1 2 #= = -t t .

Nous pouvons créer une procédure quadrilatère_aléatoire et une procédure motif :

pour quadrilatère_aléatoire
//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=200-150*rand()
a1:=110-30*rand()
a2:=110-30*rand()

quadrilatere l1 l2 l3 a1 a2
fin

pour motif
rep 10 (quadrilatère_aléatoire)
fin

Quelques ajustements sont encore nécessaires. En testant la procédure ci-dessus, nous pouvons nous 
apercevoir que certains quadrilatères sont croisés, ce qui n’est pas le cas dans l’œuvre. Afin d’exclure ce 
cas, il suffit de conditionner la création du quadrilatère à la contrainte angulaire suivante : a a 1801 2 G+t t .
Cela nous donne :

 
pour quadrilatère_aléatoire
//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=200-150*rand()
a1:=110-30*rand()
a2:=110-30*rand()

si (a1+a2<=180) alors (quadrilatere l1 l2 l3 a1 a2) sinon (quadrilatère_aléatoire)
fin

Afin de rendre plus aléatoire encore la génération des quadrilatères, nous pouvons déplacer légèrement le 
centre des quadrilatères et changer l’orientation de départ de la tortue. Nous avons besoin pour cela de 
deux variables aléatoires supplémentaires. L’angle (a_c) qui détermine l’orientation varie de 0 à 360, le 
déplacement (d_c) par rapport au centre est compris entre -50 et 50. 

Cela nous donne :

pour quadrilatère_aléatoire
//variables de positionnement aléatoire du centre
a_c:=360*rand()
d_c:= 50-100*rand()

//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=200-150*rand()
a1:=110-30*rand()
a2:=110-30*rand()

si (a1+a2<=180) alors (
(lc;td a_c;av d_c;bc;quadrilatere l1 l2 l3 a1 a2;lc;re d_c;tg a_c;bc) sinon (quadrilatère_aléatoire)
fin

Choix des couleurs du motif

Dans l’œuvre, il y a sept couleurs. Chacune d’elles correspond à une combinaison du maximum (255) dans 
la palette RVB. Ainsi : 
 

le rouge correspond à 
le vert correspond à 
le bleu correspond à 

le jaune correspond à 
le violet correspond à 

le bleu ciel correspond à 
le blanc correspond à 

RVB(255,0,0)
RVB(0,255,0)
RVB(0,0,255)
RVB(255,255,0)
RVB(255,0,255)
RVB(0,255,255)
RVB(255,255,255)

L’attribution aléatoire d’une couleur au motif peut se faire en utilisant la fonction alea(7) qui renvoie un 
entier entre 1 et 7. La procédure motif devient donc :

Motif sans déplace-
ment du centre

Motif avec un dé-
placement vertical 
du centre (d_c)

Motif avec un dé-
placement vertical 
(d_v) et une rota-
tion (a_c) du centre

pour motif
a:=alea(7)
si (a==1) alors (palette RVB 255 0 0)//rouge
si (a==2) alors (palette RVB 0 255 0)//vert
si (a==3) alors (palette RVB 0 0 255)//bleu
si (a==4) alors (palette RVB 255 255 0)//jaune
si (a==5) alors (palette RVB 255 255 255)//blanc
si (a==6) alors (palette RVB 255 0 255)//violet
si (a==7) alors (palette RVB 0 255 255)//bleu ciel

rep 10 (quadrilatère_aléatoire)
fin

Création du tableau

Les différents motifs sont répartis sur une grille de 6 par 6. Plusieurs solutions de programmation sont pos-
sibles. Nous pouvons choisir de faire une seule procédure grille en utilisant la fonction boucle (correspondant 
à une boucle for) ou bien créer une procédure qui fabriquera une ligne puis une procédure qui fabriquera la 
grille. C’est la 2ème solution qui est proposée ci-dessous.

Pour fabriquer la grille, nous choisissons un écart de 170 entre les motifs. Il faut de plus créer un carré noir 
qui fera office de fond. Cela nous donne :

pour ligne
rep 6 (motif;lc;td 90;av 170;tg 90;bc)
lc;td 90;re 6*170;tg 90;bc
fin

pour grille
palette RVB 0 0 0
remplis rep 4 (av 6*220;td 90)
lc; av 250;td 90;av 250;tg 90
rep 6 (ligne;lc;av 170)
lc;re 6*170;bc
fin

Avec les élèves

Il peut être intéressant que les élèves découvrent par eux-même le théorème d’Al-Kashi. On peut leur proposer 
de fermer un triangle à l’aide de GéoTortue en présentant cela comme une étape dans l’étude du quadrilatère.

Le problème étant posé et donnant souvent lieu à un blocage, on peut leur demander de fermer un triangle 
rectangle (théorème de Pythagore et trigonométrie) puis leur proposer la figure suivante afin de découvrir la 
relation d’Al-Kashi :

Enfin, on pourra revenir sur GéoTortue afin d’écrire une procédure triangle.

Un accompagnement particulier doit être réalisé lors de la découverte de la fonction alea(n). Les élèves 
doivent voir concrètement quel est son effet, par exemple, en éxécutant de nombreuses fois la commande 
suivante : aff alea(7). 

De la même manière, lors de la mise en place des variables aléatoires, un travail particulier est nécessaire sur 
l’utilisation de la fonction rand() et sur la manière de choisir des nombres dans un intervalle choisi.
Il faudra travailler sur les encadrements et les inégalités en montrant par exemple que :

Ce travail est réalisable au lycée. Il peut être proposé dans le cadre d’un projet et/ou de l’apprentissage de 
l’algorithmique. Cette activité demande du temps ou tout au moins une certaine habitude de GéoTortue. 

Il est aussi nécessaire d’accompagner cette activité d’un travail de recherche sur l’artiste, sur le mouvement 
dans lequel il s’inscrit, sur l’œuvre elle-même (sa date de réalisation, ses dimensions, le support utilisé ...), 
d’insister sur la différence entre une reproduction à l’ordinateur et le tableau lui-même.

Pour aller plus loin

La figure obtenue avec ce programme n’est pas celle réalisée par Véra Molnar puisque notre construction fait 
appel à plusieurs paramètres aléatoires. Ainsi, à chaque exécution du programme nous obtenons une figure 
différente. Pourquoi avoir choisi cette configuration parmi les très nombreuses possibles ? La réponse appar-
tient à l’artiste...

D’autres réalisations sont possibles à partir des procédures écrites pour reproduire cette œuvre. Véra Molnar 
a réalisé beaucoup d’œuvres à base de quadrilatères quelconques sur d’autres supports que l’ordinateur. Un 
tableau, lui aussi nommé Structure de quadrilatères, fait l’objet d’une étude similaire disponible sur le site de 
l’IREM Paris Nord.

Sources et liens

 − site officiel de Véra Molnar
 − site du digital art Museum
 − brochure de l’exposition dynamo du Grand 

Palais (Paris)
 − site de la galerie ONIRIS (Rennes)

 − site de l’IREM Paris Nord
 − site de GéoTortue


